

38

SEXUAL SELECTION

In collaboration with Shelley Ball

Objectives

- Determine how female choice affects allele and genotype frequencies in a population.
- Determine how initial allele frequencies influence the evolution of allele frequencies through female choice.
- Evaluate how natural selection can counter sexual selection in the evolution of a trait.

Suggested Preliminary Exercises: Hardy-Weinberg Equilibrium and Multilocus Hardy Weinberg

INTRODUCTION


From a genetic perspective, evolution is often described as a change in allele frequency over time. What mechanisms cause changes in allele frequencies? Gene flow, mutations, and genetic drift can all spur such change. **Natural selection**—the differential survival and reproductive success of individuals in populations—is another major evolutionary force. Natural selection simply means that if some individuals have genetic characteristics that are well-suited for a particular environment, they will on average survive better and produce more offspring than other individuals in the population, thereby changing allele frequencies in subsequent generations.

In some cases, natural selection arises from differences in mating success: certain individuals possess traits that cause them to be perceived as “better” mates, and hence to mate more frequently than other individuals in the population. For example, the long, bright tails of male peacocks may have evolved because females preferentially selected males with the longest and brightest tails (the selective force was female choice). This difference in mating success due to such traits is called **sexual selection**.

Charles Darwin thought that sexual selection was different from natural selection, saying “Sexual selection ... depends not on a struggle for existence, but on a struggle between the males for possession of the females; the result is not death to the unsuccessful competitor, but few or no offspring” (Darwin 1871).

The theory of sexual selection assumes the selection of traits that are purely concerned with maximizing mating success. Males can “increase the odds” of mating by having traits (such as the long, bright tail feathers in male peacocks) that attract females. Males can also maximize their mating success by the “brute force” method:

Figure 1 When some females prefer males with long tails, males with the T_2 genotypes will increase in frequency in the population in the next generation (bottom). P_1 females randomly choose to mate with both long- and short-tailed males, while P_2 females prefer males with long tails. If this preference is strong enough, and if P_2 females are sufficiently frequent in the population, long-tailed males may mate more successfully on average and thus produce more offspring than short-tailed males. These offspring will tend to inherit both the allele for long tails (from their male parents) and the allele for tail preference (from their female parents), so that as selection increases the frequency of T_2 it also increases the frequency of P_2 . As P_2 becomes more frequent and an increasing proportion of females favors long tails, the advantage of having a long tail increases. Alleles T_2 and P_2 may thus both increase in frequency over time, at ever-increasing rates. The change in genotype frequency over time for males is shown in Figure 2. Note that male genotypes containing a T_2 allele increase in frequency, while male genotypes with T_1 decrease in frequency. (From Futuyma 1998.)



outcompeting other males for mating opportunities (male-male contests). Female traits may not be so visible; females maximize their fitness by selecting males that somehow enhance their own fitness or the fitness of their offspring. A female might select males that have “good genes” which enhance her offspring’s fitness (an indirect benefit of mate choice), or by selecting males that are “good parents/mates,” which enhance the female’s own survival and reproductive success (a direct benefit of mate choice).

In these cases it’s fairly easy to imagine how females that choose beneficial mates can be favored in a population, and how such choices influence the evolution of a species (Alcock 2001). But what happens when there is no direct or indirect fitness benefit associated with mate choice? Can a population still evolve due to sexual selection? The answer, in theory, is yes. Ronald Fisher introduced the theoretical argument in 1930. Fisher realized that sexual selection could cause populations to evolve when there is no fitness gain associated with mate choice, and that sometimes even traits that decrease survivorship, such as an extraordinarily long tail, can evolve in a population as a result. Fisher’s model is called **runaway sexual selection**.

An important underlying assumption of Fisher’s model is that both the female preference and the male trait (i.e. tail length) must be under genetic control. (Remember, traits cannot evolve unless they have a genetic basis.) So, let’s imagine that males have a gene associated with tail length in which males have either a T_1 (short) or a T_2 (long) genotype (let’s assume, for simplicity, that we’re dealing with a haploid organism). Females also have these genes for tail length but do not express them. Let’s further assume that the T_2 genotype has a fitness cost—perhaps males with long tails have higher mortality rates because predators capture them more easily. Let’s also imagine that a separate, *nonlinked* gene determines mating preference, where the genotype P_1 indicates no preference for tail length but the P_2 allele indicates a strong preference for long tails. Both males and females have the P gene, but only females express the gene when they

solicit matings. Thus, both sexes carry an allele for both the P and T genes. Because of this, selection for a particular allele of one gene can “drag along” a particular allele of the other gene. This association leads to a genetic correlation between the tail length gene and the mating preference gene, as shown in Figure 1.




Figure 2

If runaway sexual selection actually happens in nature, why don't we commonly see birds with super-long tails? Although sexual selection for long tails increases the fitness of long-tailed males over short-tailed males, natural selection may select against long-tailed males through decreased survivorship. For example, if a tail is so long that the bird has troubles escaping from predators, there will be fewer long-tailed males in the population. Depending on the strength of selection against long-tailed males, we can expect some equilibrium level that would balance survival costs of having a long tail with the reproductive benefits of having such a tail. Figure 3 shows an example of how natural selection can drive the T_2 allele to extinction by substantially decreasing the survival probability of T_2 males.




Figure 3

In Fisher’s (1958) model of sexual selection, he assumed that the female’s preference must confer some sort of selective advantage and that this advantage was necessary to “get the ball rolling” in the runaway process. However, later work by Kirkpatrick (1982) showed that an initial selective advantage was not a necessary prerequisite for the runaway process of sexual selection and that evolution of the male trait could occur without selection for or against female preference. We will model Fisher’s runaway process of sexual selection and in doing so, show that no initial selective advantage of female preference is necessary for generating the runaway process.

PROCEDURES

In this exercise, you’ll set up a runaway sexual selection model and see first-hand how the runaway process works. You’ll model a population of 2000 haploid individuals (1000 males and 1000 females). There will be two alleles, *P* and *T*, as previously described, and thus there are 4 possible genotypes: P_1T_1 , P_1T_2 , P_2T_1 , and P_2T_2 . You will set up a table of mating preferences that indicate the preferences of a female genotype for the various male genotypes. These mating preferences will be converted to mate selection probabilities that account for the frequencies of male genotypes in the population. Once the mating preferences are assigned, you will simulate the reproduction of offspring as a diploid organism (by joining the male and female partner’s genotypes), and then will simulate meiosis to so that organisms so that the haploid system is maintained. Once the offspring are generated, you will compute the numbers of P_1T_1 , P_1T_2 , P_2T_1 , and P_2T_2 individuals in the offspring population, and then compute their allele frequencies. And finally, you will develop a macro to track the allele frequencies over time to see how the various genotypes evolve.

Admittedly, this is a pretty complicated spreadsheet, so take your time as you work through it and try to keep the bigger picture in mind as you develop the model. As always, save your work frequently to disk.

INSTRUCTIONS

A. Set up the model parental population.

1. Open a new spreadsheet and set up headings as shown in Figure 4.

2. In cells C5–C8 and E5–E8, enter the number of individuals with each genotype.

ANNOTATION

We will start by setting up a parent population that contains 1000 males and 1000 females. The tail length locus, *T*, has 2 alleles, T_1 (short tail) and T_2 (long tail). The preference locus, *P*, has two alleles, P_1 (no preference for tail length) and P_2 (preference for long tails).

	A	B	C	D	E	F
1	Sexual Selection					
2			PARENTAL GENOTYPE NUMBERS			
3				Tally		Tally
4	Genotype		Male	0	Female	0
5	T1P1	<i>Short tail, no preference</i>	250		250	
6	T1P2	<i>Short tail, preference</i>	250		250	
7	T2P1	<i>Long tail, no preference</i>	250		250	
8	T2P2	<i>Long tail, preference</i>	250		250	

Figure 4

The possible genotypes for the population are given in cells A5–A8.

Enter 250 in cells C5–C8 and E5–E8.

To begin, we will assume that the all genotypes are equally represented in the population. You will be able to change these cells later in the exercise.

3. Enter 0 in cells D4 and F4.

4. In cell D5–D8, enter formula to tally the male genotypes.

5. Set up the tally for females in cells F5–F8.

6. Set up new spreadsheet headings as shown in Figure 5.

7. Set up a linear series from 0 to 999 in cells A22–A1021.

8. In cells B22–B1021, use the **LOOKUP** function to assign genotypes to the males.

9. In cells C22–C1021, enter a formula to assign genotypes to the females.

10. Save your work.

B. Set up the mating preferences and mate selection probabilities.

1. Set up new column headings as shown in Figure 6.

2. Enter the female mate selection preferences shown in cells I5–L8.

This is a “place holder” to tally the total number of males and females in cells D5–D8 and cells F5–F8. It is necessary so that we can assign genotypes properly to the 1000 males and females.

Enter **=C5** in cell D5.

Enter the formula **=D5+C6** in cell D6. Copy this formula down to cell D8.

This is a running tally that counts the total number of individuals as we consider additional genotypes. The final result in cell D8 should be 1000.

Enter **=E5** in cell F5.

Enter **=F5+E6** in cell F6. Copy the formula down to cell F8. Your total should be 1000 in cell F8.

	A	B	C
19	Initial population		
20		Male adult	Female adult
21	Individual	genotype	genotype

Figure 5

Enter 0 in cell A22.

In cell A23, enter **=1+A22**. Copy your formula down to cell A1021. This assigns a number to each male and each female in the population.

Enter the formula **=LOOKUP(A22,\$D\$4:\$D\$8,\$A\$5:\$A\$8)**. Copy this formula down to cell B1021.

The **LOOKUP** function looks up a value (A22) in a vector that you specify (cells **\$D\$4:\$D\$8**), and returns a genotype for the individual given in the vector **\$A\$5:\$A\$8**. (A vector is a single row or column of values). The result of this function is that genotypes are assigned to individuals in exactly the numbers that you specified in cells C5–C8.

Examine your first 10 genotypes. They should all be T_1P_1 . To see how the function works, change cell C5 to 1. Now examine the genotypes of your first 10 individuals. The first male should be T_1P_1 , but the rest of the males should be T_1P_2 . When you feel you have a handle on how this function works, return cell C5 to 250.

Enter the formula **=LOOKUP(A22,\$F\$4:\$F\$8,\$A\$5:\$A\$8)** in cell C22. Copy your formula down to cell C1021.

The formula for females works in the same way as that for males, using the female tallies.

	G	H	I	J	K	L
2	MATE SELECTION PREFERENCES					
3			Male genotype			
4			T1P1	T1P2	T2P1	T2P2
5		T1P1	0.25	0.25	0.25	0.25
6	Female	T1P2	0	0	0.5	0.5
7	genotype	T2P1	0.25	0.25	0.25	0.25
8		T2P2	0	0	0.5	0.5
9						
10	MATE SELECTION PROBABILITIES					
11			Male genotype			
12			T1P1	T1P2	T2P1	T2P2
13		Survival =>	1	1	1	1
14		T1P1				
15	Female	T1P2				
16	genotype	T2P1				
17		T2P2				

Figure 6

Cells H5–H8 represent the genotypes of females, and cells I4–L4 represent the genotype of a female’s potential mate. The entries in cells I5–L8 establish the female’s mating preferences. Thus, a female with genotype T_1P_1 has the “no preference for tail length gene,” so her preferences are identical for all four male genotypes. A female with genotype T_1P_2 or T_2P_2 has the P_2 “preference for long tailed males gene,” so she will prefer to mate with males that have a genotype T_2P_1 or T_2P_2 , but will not prefer males with genotypes T_1P_1 or T_1P_2 . **Note that the probabilities in each row in this table must sum to 1!**

3. In cell I14, enter a formula to compute the probability that a mating between the specified genotypes will take place.

Enter the formula $=I5*C5/(I5*C5+J5*C6+K5*C7+L5*C8)$ in cell I14.

Although female mating preferences have been established, mating probabilities must also consider the number of males of each genotype in the population. The formula in cell I14 makes this adjustment and computes the probability that a T_1P_1 female will mate with a T_1P_1 male. The formula multiplies the preference for T_1P_1 males by the number of T_1P_1 males in the population, then adjusts this result by dividing by preference \times number for all of the genotypes in the population.

4. Enter formulae to compute the remaining mate selection probabilities.

Double-check your formulae against Figure 7.

	I	J
14	$=I5*C5/(I5*C5+J5*C6+K5*C7+L5*C8)$	$=J5*C6/(I5*C5+J5*C6+K5*C7+L5*C8)$
15	$=I6*C5/(I6*C5+J6*C6+K6*C7+L6*C8)$	$=J6*C6/(I6*C5+J6*C6+K6*C7+L6*C8)$
16	$=I7*C5/(I7*C5+J7*C6+K7*C7+L7*C8)$	$=J7*C6/(I7*C5+J7*C6+K7*C7+L7*C8)$
17	$=I8*C5/(I8*C5+J8*C6+K8*C7+L8*C8)$	$=J8*C6/(I8*C5+J8*C6+K8*C7+L8*C8)$

	K	L
14	$=K5*C7/(I5*C5+J5*C6+K5*C7+L5*C8)$	$=L5*C8/(I5*C5+J5*C6+K5*C7+L5*C8)$
15	$=K6*C7/(I6*C5+J6*C6+K6*C7+L6*C8)$	$=L6*C8/(I6*C5+J6*C6+K6*C7+L6*C8)$
16	$=K7*C7/(I7*C5+J7*C6+K7*C7+L7*C8)$	$=L7*C8/(I7*C5+J7*C6+K7*C7+L7*C8)$
17	$=K8*C7/(I8*C5+J8*C6+K8*C7+L8*C8)$	$=L8*C8/(I8*C5+J8*C6+K8*C7+L8*C8)$

Figure 7

Double-check your results as well. Since there are currently equal numbers of male genotypes in the population, the mate selection probabilities should be the same as the mate selection preferences (Figure 8).

	G	H	I	J	K	L
10	MATE SELECTION PROBABILITIES					
11			Male genotype			
12			T1P1	T1P2	T2P1	T2P2
13		Survival =>	1	1	1	1
14		T1P1	0.25	0.25	0.25	0.25
15	Female	T1P2	0	0	0.5	0.5
16	genotype	T2P1	0.25	0.25	0.25	0.25
17		T2P2	0	0	0.5	0.5

Figure 8

5. Enter a survival probability for males in cells I13–L13.

6. Save your work.

C. Simulate parental matings.

1. Set up new headings as shown in Figure 9.

Enter the number 1 in cells I13–L13.

Currently the survival probability is set to 1 so that all male genotypes have equal survival. Later in the exercise, you will be able to change these values so that males with long tails have a lower probability of survival.

Our goal is to have the spreadsheet look up the genotype of female parents (in column C) and match their genotype to genotypes listed in cells H14–H17. Ultimately, we want to determine the genotype of the female's selected mate, listed in cells I12–L12. To choose mates according to the probabilities given, we will use four different functions: **MATCH**, **INDEX**, **RAND**, and **IF**. The combination of these formulae will allow us to generate the genotype of a mate for each female in the population in column J.

	D	E	F	G	H	I	J
19	Mate "selection"						
20		T1P1	T1P2	T2P1	T2P2	Random	Selected
21	Match	index	index	index	index	number	male mate

Figure 9

2. In cell D22, enter the formula **=MATCH(C22,\$H\$14:\$H\$17)**.

3. In cell E22, enter the formula **=INDEX(\$H\$14:\$L\$17,D22,2)**.

The **MATCH** formula returns the *relative position* of an item in a table that matches the condition you specify. The **MATCH** function has the syntax **MATCH (lookup_value,lookup_array,match_type)**, where **lookup_value** is the value you use to find the value you want in a table, **lookup_array** is a contiguous range of cells containing possible lookup values, and **match_type** tells the spreadsheet how to match the lookup value to the lookup array (by not specifying match-type, the default is used). In cell D22, the formula **=MATCH(C22,\$H\$14:\$H\$17)** tells the spreadsheet to find the genotype listed in cell C22, and return the *relative position* of that genotype in the \$H14–\$H17 table. For example, the genotype of female 1 in the spreadsheet is T_1P_1 . Excel returns the value 1 to indicate that T_1P_1 individuals occupy the first position in our table. If female 1 had the genotype T_2P_2 , the program would return the number 4 (fourth position).

The **INDEX** formula returns the *value* of an element in a table, once you identify the row and column number that should be returned. The **INDEX** formula has the syntax: **INDEX(array,row_num,column_num)**, where **array** is a range of cells in a table; **row_num** selects the row in the table from which to return a value, and **column_num**

4. Use the **INDEX** function to index the T_1P_2 , T_2P_1 , and T_2P_2 genotypes in cells F22–H22.

5. In cell I22, use the **RAND** function to generate a random number between 0 and 1.

6. In cell J22, enter a formula to establish the genotype of that female's selected mate.

7. Select cells D22–J22, and copy and the formula down to row 1021

8. Save your work.

D. Impose natural selection on males.

1. Set up new headings as shown in Figure 10.

2. In cells K22–K1021 enter a formula to compute which males survive to breed and produce offspring.

selects the column in table from which to return a value. In cell E22, the formula **=INDEX(\$H\$14:\$L\$17,D22,2)** tells the spreadsheet to examine the range of cells in H14–L17 and go to the row designated in cell D22 (derived from your **MATCH** formula) and column 2 (which indicates the probability of mating with an T_1P_1 male). The spreadsheet will then return the value associated with this row and column intersection. Your result should be 0.25.

Enter the formula **=INDEX(\$H\$14:\$L\$17,D22,3)** in cell F22.

Enter the formula **=INDEX(\$H\$14:\$L\$17,D22,4)** in cell G22.

Enter the formula **=INDEX(\$H\$14:\$L\$17,D22,5)** in cell H22.

These four formulae simply generate the appropriate mating probabilities for each individual in the population.

Enter **=RAND()** in cell I22.

This formula randomly determines the genotype of the mate for each individual in the population. When you press F9, the calculate key, you generate a new set of random numbers.

Enter the formula **=IF(I22<=E22,\$I\$12,(IF(I22<=E22+F22,\$J\$12,(IF(I22<=E22+F22+G22,\$K\$12,\$L\$12)))))** in cell J22.

The formula in cell J22 looks complicated but really it's not. The formula is simply four nested **IF** statements. The formula tells the spreadsheet to examine cell **I22** (the random number). If **I22** is less than or equal to the value in cell **E22** (**<=E22**), then return the genotype in cell **\$I\$12**; otherwise walk through the next **IF** statement. The next statement examines cell **I22**, and if its value is less than or equal to the values in cells **E22 + F22** (**<=E22+F22**), then return the genotype in cell **\$J\$12**; otherwise walk through the third **IF** statement. The third statement examines cell **I22**, and if its value is less than or equal to the sum of **E22, F22, and G22** (**<=E22+F22+G22**), return the genotype in cell **\$K\$12**, otherwise return the value in cell **\$L\$12**.

This will establish the selected mate's genotype for each female in the population. Review your spreadsheet entries and results for the first five individuals and make sure you understand how mates were determined for the females.

We set up the spreadsheet so that selection against a particular genotype occurs after female mating probabilities have been established. Thus, selection against a genotype does not influence the mating probabilities themselves. For now, each genotype has a survival probability of 1 (given in cells I13–L13), indicating that there is no "cost" to having a long tail. If we wished to impose selection against long-tailed males, we would alter the survival probabilities in cells I13–L13.

	K
20	Natural
21	selection

Figure 10

Enter the formula **=IF(RAND()<HLOOKUP(J22,\$I\$12:\$L\$13,2),J22,"")** in cell K22. Copy the formula down to cell K1021.

The formula uses an **HLOOKUP** function to find the genotype of the selected mate for female 1 (J22) in the table of cells I12–L13, and finds that male's survival probability in the second row of the table. The **RAND()** function draws a random number between 0 and 1. The **IF** function determines whether this random number is less than the appropriate survival probability. If the random number is *less than* the survival probability,

3. Save your work.

E. Establish offspring genotypes and allele frequencies.

1. Set up new column headings as shown in Figure 11.

2. In cell L22, enter a formula to combine the female's haploid genotype with her mate's haploid genotype to produce a diploid offspring (only if the male survived to breed).

3. In cells M22 and N22, use the **IF**, **RAND**, and **MID** functions to generate the genotypes of haploid individuals.

4. Select cells L22–N22 and copy their formulae down to row 1021.

5. Set up new column headings as shown in Figure 12.

the male lives and his genotype (J22) is returned. If the random number is *greater than* the survival probability, the male dies and a period (".") is returned, indicating a death.

	L	M	N
19	Offspring population		
20	Diploid	Male haploid	Female haploid
21	genotype	genotype	genotype

Figure 11

Enter the formula `=IF(K22=".",",",C22&J22)` in cell L22.

If the male in cell K22 is dead, the formula returns a missing value (.). If the male is not dead, the spreadsheet returns the combination of cells C22 and J22; the **&** function simply concatenates the two cells.

Enter the formula `=IF(RAND()<0.5,MID(L22,1,2),MID(L22,5,2))`

&IF(RAND()<0.5,MID(L22,3,2),MID(L22,7,2)) in cells M22 and N22.

Our goal is to generate male and female offspring that have a single *T* allele and a single *P* allele. We'll let meiosis occur with random segregation of alleles.

The **MID** function returns a specific number of characters from a text string, starting at the position you specify, and based on the number of characters you specify. It has the syntax **MID(text,start_num,num_chars)**. The first part of the formula, **IF(RAND()<0.5,MID(L22,1,2),MID(L22,5,2))**, tells the spreadsheet to draw a random number between 0 and 1. If that random number is <0.5, return the value associated with **MID(L22,1,2)**, otherwise returns the value associated with **MID(L22,5,2)**. The **MID(L22,1,2)** portion of the formula tells the spreadsheet to examine cell L22 and, starting at the first character, return 2 characters. The **MID(L22,5,2)** portion of the formula examines cell L22, and starting at the fifth character, returns 2 characters. This portion of the formula returns a randomly selected *T* allele. The second **IF** statement is analogous and randomly selects the *P* allele for each offspring. The two alleles are joined by the **&** symbol.

	A	B	C	D	E	F
9			OFFSPRING			
10			Genotype numbers		Genotype frequencies	
11			Male	Female	Male	Female
12	T1P1	Short tail, no preference				
13	T1P2	Short tail, preference				
14	T2P1	Long tail, no preference				
15	T2P2	Long tail, preference				
16						

Figure 12

6. In cells C12–C15 and D12–D15, use the **COUNTIF** function to count the number of offspring male and female genotypes, respectively. Sum the totals in cells C16 and D16.

Double-check your formulae against Figure 13.

	C	D
10	Genotype numbers	
11	Male	Female
12	=COUNTIF(\$M\$22:\$M\$1021,A12)	=COUNTIF(\$N\$22:\$N\$1021,A12)
13	=COUNTIF(\$M\$22:\$M\$1021,A13)	=COUNTIF(\$N\$22:\$N\$1021,A13)
14	=COUNTIF(\$M\$22:\$M\$1021,A14)	=COUNTIF(\$N\$22:\$N\$1021,A14)
15	=COUNTIF(\$M\$22:\$M\$1021,A15)	=COUNTIF(\$N\$22:\$N\$1021,A15)
16	=SUM(C12:C15)	=SUM(D12:D15)

Figure 13

7. In cells E12–F16, compute the male and female offspring genotype frequencies.

Double-check your formulae against Figure 14.

	E	F
10	Genotype frequencies	
11	Male	Female
12	=C12/\$C\$16	=D12/\$D\$16
13	=C13/\$C\$16	=D13/\$D\$16
14	=C14/\$C\$16	=D14/\$D\$16
15	=C15/\$C\$16	=D15/\$D\$16
16	=SUM(E12:E15)	=SUM(F12:F15)

Figure 14

8. Save your work.

F. Track genotype frequencies over time.

1. Set up new headings as shown in Figure 15, but extend your generations to 15.

	N	O	P	Q	R	S	T	U	V
2		MALES				FEMALES			
3	Generation	T1P1	T1P2	T2P1	T2P2	T1P1	T1P2	T2P1	T2P2
4	1								
5	2								
6	3								
7	4								
8	5								

Figure 15

2. Open Tools | Options | Calculation and set the calculation key to manual.

3. Write a macro to track allele frequencies over time.

The macro needs to perform the following steps:

- Paste the genotype numbers of the offspring into the parental population cells.
- Press the calculate key to simulate mate selection, natural selection, and breeding.
- Record the offspring’s allele frequencies in the cells O4–V18.

There are many ways to write a macro to conduct these steps. We suggest one way, but you may see other (perhaps easier) steps. Put your macro function in the “record macro” mode and assign a shortcut key (see Exercise 2). Record the following operations:

- Press F9, the calculate key, to generate new random numbers (and hence new matings and offspring for the parental population).
- Select cells E12–E15. Copy.
- Select cell O3.
- Open Edit | Find. The dialog box in Figure 16 will appear. Leave the Find What box blank and Search By Columns. Select Find Next and then Close. Your cursor should move down to the next blank cell.




Figure 16

- Open Edit | Paste Special and select the Paste Values and Transpose options. Click OK.
- Select cells F12–F15. Copy.
- Select cell S3.
- Open Edit | Find.
- Click on Find Next and then Close.
- Open Edit | Paste Special, and select the Paste Values and Transpose options.
- Select cells C12–C15. Copy.
- Select cell C5.
- Open Edit | Paste Special and select the Paste Values option.
- Select cells D12–D15. Copy.
- Select cell E5.
- Open Edit | Paste Special and select the Paste Values option.

Stop recording. Now when you press your shortcut key, your macro will record the allele frequencies of the various genotypes for males and females.

4. Run the macro 15 times (i.e., over 15 generations).

5. Save your work.

G. Create graphs.

1. Graph the allele frequencies of males and females over time. Make a separate graph for each sex.

Use the line graph option and label your axes fully. Your graphs should resemble Figures 17 and 18.




Figure 17




Figure 18

2. Save your work.

QUESTIONS

- Interpret your model results. For each sex, explain how the genotypes (T_1P_1 , T_1P_2 , T_2P_1 , T_2P_2) evolve (change in frequency) from one generation to the next. Which genotypes went extinct; which genotypes persisted? Did this differ for males and females? If so, why? What mechanism allows for the evolution of the T and P alleles?
- In your model, females with the P_2 allele mate only with the T_2 males—no exceptions. In reality, perhaps not all females will be able to mate with T_2 males, and so some P_2 females will mate with T_1 males. Change the choice parameters

in cells I6–L6 and I8–L8 to 0.1, 0.1, 0.4, and 0.4. Reset your genotype numbers in cells C5–C8 and E5–E8 to 250. Clear your old macro results, and run your macro again. How does the “strength” of sexual selection influence the change in allele frequencies from one generation to the next?

3. How does natural selection influence the evolution of the T_2 trait even when P_2 females have full preference for T_2 male? Return the mate selection preferences to their original values as shown:

	G	H	I	J	K	L
2	MATE SELECTION PREFERENCES					
3			Male genotype			
4			T1P1	T1P2	T2P1	T2P2
5		T1P1	0.25	0.25	0.25	0.25
6	Female	T1P2	0	0	0.5	0.5
7	genotype	T2P1	0.25	0.25	0.25	0.25
8		T2P2	0	0	0.5	0.5

Decrease the survival probabilities of T_2 males in cells K13–L13 by increments of 0.1. With each incremental decrease in survival, run your macro again (clear your old results, and make sure to reset the initial genotype numbers to 250 in cells C5–C8 and E5–E8). What level of natural selection “puts the brakes” on sexual selection?

4. How do starting allele frequencies affect the outcome of a simulation? The initial genotypes we used to build the spreadsheet are admittedly very unusual; before sexual selection for tail length begins, it is much more likely that at least one (if not both) of the alleles T_2 and P_2 will be new and very rare mutations. That is, either there will be variety in tail length (long and short tails both occur with some regularity) when a mutation causes one female (or a few sisters) to prefer long tails, *or* there will already be a preference for a trait that does not yet exist, and mutation will create that trait in one male (or a few brothers). We can use this spreadsheet model to test both of these initial conditions.

Set the genotype survivals back to 1 and set the initial genotype numbers as shown below:

	A	B	C	D	E	F
3				Tally		Tally
4	Genotype		Male	0	Female	0
5	T1P1	<i>Short tail, no preference</i>	500	500	500	500
6	T1P2	<i>Short tail, preference</i>	0	500	0	500
7	T2P1	<i>Long tail, no preference</i>	500	1000	490	990
8	T2P2	<i>Long tail, preference</i>	0	1000	10	1000

In this population, half the males have long tails and half have short, but none of the males carry the allele for preferring long tails. Approximately half the females carry the allele for long tails, but by some unusual chance, 10 sisters in this generation all received a mutated gene that causes them to mate exclusively with long-tailed males. Clear your previous results from cells O4–V18 and run your macro to see what happens to genotype frequencies over 15 generations. Do these initial conditions result in runaway sexual selection?

5. Set the initial genotype frequencies to those shown below:

	A	B	C	D	E	F
2			PARENTAL GENOTYPE NUMBERS			
3				Tally		Tally
4	Genotype		Male	0	Female	0
5	T1P1	<i>Short tail, no preference</i>	900	900	900	900
6	T1P2	<i>Short tail, preference</i>	99	999	100	1000
7	T2P1	<i>Long tail, no preference</i>	0	999	0	1000
8	T2P2	<i>Long tail, preference</i>	1	1000	0	1000

In this population, 10% of the females would prefer to mate with a long-tailed male, although almost the entire population consists of short-tailed males. Approximately 10% of the males also carry the allele for preference, even though they do not express it, but almost all the males have short tails. One lone male has mutated to have a tail that is longer than the others.

To make these initial conditions a little more plausible, we need to allow that the P_2 allele does not confer *absolute* preference—otherwise the females that carried it up to this generation would not have mated, and the allele would have been lost. Resetting the mate selection preferences as shown below will give us females who would strongly prefer long-tailed males but who will settle for short-tailed males in a pinch.

	G	H	I	J	K	L
2	MATE SELECTION PREFERENCES					
3			Male genotype			
4			T1P1	T1P2	T2P1	T2P2
5		T1P1	0.25	0.25	0.25	0.25
6	Female	T1P2	0.01	0.01	0.49	0.49
7	genotype	T2P1	0.25	0.25	0.25	0.25
8		T2P2	0.01	0.01	0.49	0.49

Clear your previous results from cells O4–V18 and run your macro to see what happens to genotype frequencies over 15 generations. Do these initial conditions result in runaway sexual selection?

6. Can runaway sexual selection occur when $P_2 = 0$? Set your initial conditions so that all females and 995 males in the population have the genotype T_1P_1 and 5 males have the genotype T_2P_1 . Then set the mate selection preferences as shown below:

	G	H	I	J	K	L
2	MATE SELECTION PREFERENCES					
3			Male genotype			
4			T1P1	T1P2	T2P1	T2P2
5		T1P1	0.3	0	0.7	0
6	Female	T1P2	0	0	0	0
7	genotype	T2P1	0.3	0	0.7	0
8		T2P2	0	0	0	0

Clear your previous results from cells O4–V18 and run your macro to see what happens to genotype frequencies over 15 generations. Analyze your model outputs and explain how this might occur in nature.

7. Genetic drift can also influence changes in allele frequencies over time. To examine the effects of genetic drift on this model of sexual selection, we are going to manipulate the population size by running the model using different initial genotype numbers. For example, instead of having 1000 individuals of each sex, start with 500 individuals of each. As with our initial conditions, simply start with equal numbers of each genotype (so in this case, each genotype number will be 125). Clear your results from your last simulation, set the mate selection preferences in cells I5–L8 back to their initial values, and run your model. Record your results and then run the model a few more times, each time changing the total genotype numbers (but make sure there are equal numbers of each genotype). What effect does changing the population size have on the outcome of the model?

LITERATURE CITED

- Alcock, J. 2001. The evolution of reproductive success. Chapter 11, pp. 316–359 in *Animal Behavior: An Evolutionary Approach*, 7th Edition. Sinauer Associates, Sunderland, MA.
- Basolo, A. L. 1990. Female preference predates the evolution of the sword in sword-tail fish. *Science* 250: 808–810.
- Darwin, C. 1871. *The Descent of Man, and Selection in Relation to Sex*. John Murray, London.
- Fisher, R. A. 1930. *The Genetical Theory of Natural Selection*. Clarendon Press, Oxford.
- Futuyma, D. 1998. *Evolutionary Biology*, 3rd Edition. Sinauer Associates, Sunderland, MA.
- Kirkpatrick, M. 1982. Sexual selection and the evolution of female choice. *Evolution* 3: 1–12.